
	Function name
	f_TC_8_1_6_1_2_1_NR5GC_TestBody

	Reason for change
	3. According to 38.331 when logMeasReport is available then the UE information response message shall be received via SRB2
According to 38.331 sec 5.7.10.3
   1> if the logMeasReport is included in the UEInformationResponse:
   2> submit the UEInformationResponse message to lower layers for transmission via SRB2;
in this test case the logMeasReport is included and therefore it has to be sent on SRB2

4. According to the prose When UE information response message is received, the 
measResultServCell information will be Cell 1 (current implementation is checking against cell 11) and the 
measResultNeighCells has to be checked( current implementation is not checking this) 


	Summary of change
	3. At Step 20 modified the template to use car_NR_SRB_RrcPdu_IND and receive the message on SRB2
4. Updated the variables and Step 20 to check for the measResultServCell with Cell 1 information and also included the measResultNeighCells to be checked.

	TTCN module

Before Change:
	function f_TC_8_1_6_1_2_1_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var PLMN_Identity v_PlmnID_NRCell1 := f_NR_CellInfo_GetPLMN (nr_Cell1);
    var PLMN_Identity v_PlmnID_NRCell11 := f_NR_CellInfo_GetPLMN (nr_Cell11);
    var CellIdentity v_CellIdentity_NRCell11 := f_NR_CellInfo_GetCellIdentity(nr_Cell11);
    var TrackingAreaCode v_TAC_NRCell11 := f_NR_CellInfo_GetTAC (nr_Cell11);
    var integer i;
    var integer v_FieldCnt;
    var LogMeasInfo_r16 v_LogMeasInfo_r16;
    var template (present) MeasQuantityResults v_MeasQuantityResults;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var AbsoluteTimeInfo_r16 v_AbsoluteTimeInfo_r16 := f_MDT_GetAbsolute_TimeStamp();
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();

    //@siclog "Step 1" siclog@
    // SS transmits a LoggedMeasurementConfiguration message
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                    cs_TimingInfo_Now,
                                    cs_38508_LoggedMeasurementConfiguration(v_PlmnID_NRCell1,
                                                                            v_AbsoluteTimeInfo_r16,
                                                                            cs_LoggedMeasurementConfiguration_reportType_Periodical)));

    //@siclog "Steps 2" siclog@
    // The SS transmits an RRCRelease message.
    f_NR_RRCRelease(nr_Cell1);

    //@siclog "Step 3" siclog@
    // Wait 5s to allow UE to activate logging.
    f_Delay (5.0);

    //@siclog "Step 4" siclog@
    //Set SS/PBCH EPRE levels acc. to row "T1"
    f_NR_SetCellPower (nr_Cell1, -85, -91);
    f_NR_SetCellPower (nr_Cell11, -78, -82);

    //@siclog "Step 5" siclog@
    //The UE transmits an RRCSetupRequest message on Cell 11.
    f_NR_RRC_SetupRequest_Def(nr_Cell11);

    //@siclog "Step 6" siclog@
    // The SS transmit an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell11);

    //@siclog "Step 7" siclog@
    // Check: Does the UE include the IE logMeasAvailable in the RRCSetupComplete message?
    f_NR_RRCSetupComplete_Def(nr_Cell11,
                              cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def,-,-,-,cr_RRCSetupComplete_v1610_IEs(*,*,cr_UE_MeasurementsAvailable_r16(true_,omit,omit,omit,omit))));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.1.6.1.2.1 Step 7");

    //@siclog "Steps 8-10" siclog@
    // Steps 4 to 6a1 of generic test procedure in TS 38.508-1[4] subclause 4.9.5.2.2-1 are performed on Cell 11.
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11, v_GMM_MobilityInfo);

    //@siclog "Steps 11-18" siclog@
    // Steps 1 to 8 of the generic procedure in TS 38.508 Table 4.5.4.2-3 are executed to successfully complete the service request procedure on Cell 11.
    f_NR_Paging_Def(nr_Cell11, RRC_IDLE);
    f_NR_RRC_SetupRequest_Def(nr_Cell11);
    f_NR_RRCSetup_Def(nr_Cell11);
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell11,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def,-,-,-,cr_RRCSetupComplete_v1610_IEs(*,*,cr_UE_MeasurementsAvailable_r16(true_,omit,omit,omit,omit))));
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell11, v_ReceivedMsg);

    //@siclog "Step 19" siclog@
    // The SS transmits a UEInformationRequest message on Cell 11.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell11,
                                    -,
                                    cs_38508_UEInformationRequest(tsc_NR_RRC_TI_Def, -, true_)));

    // siclog "Step 20" siclog@
    // Check: Does the UE transmit a UEInformationResponse message with the IEs absoluteTimeStamp, traceReference, traceRecordingSessionRef and tce-Id set to same value as received in the LoggedMeasurementConfiguration message in step 1; and a LogMeasInfoList with at least two  entries with serving cell idle mode measurements and where in at least one of the entries the IE measResultListNR include a neighbouring cell measurement of Cell 11; and  where the relativeTimeStamp is increased between the subsequent LogMeasInfoList entries by at least the value of configured loggingInterval in the received LoggedMeasurementConfiguration message in step 1.
    v_MeasQuantityResults := f_NR_Get_MeasQuantityResults();
    SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell11,
                                       cr_38508_UEInformationResponse(tsc_NR_RRC_TI_Def, -,-,
                                                                      cr_LogMeasReport_r16(v_AbsoluteTimeInfo_r16,
                                                                                           cr_TraceReference_r16(v_PlmnID_NRCell1),
                                                                                           '001A'O,
                                                                                           '05'O,
                                                                                           ?,
                                                                                           omit,omit,omit
                                                                                           )))) -> value v_ReceivedAsp;

    v_FieldCnt := lengthof(v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.logMeasReport_r16.logMeasInfoList_r16);
    if (v_FieldCnt < 2) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 20: Not enough LogMeasurementInfo entries")
    }
    for (i:=0; i<v_FieldCnt; i:=i+1){
        v_LogMeasInfo_r16 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.logMeasReport_r16.logMeasInfoList_r16[i];
        if (not match(v_LogMeasInfo_r16, cr_LogMeasInfo_r16(-, cr_CGI_Info_Logging_r16(v_PlmnID_NRCell11, v_CellIdentity_NRCell11, v_TAC_NRCell11),
                                                           cr_MeasResultServCell_r16(v_MeasQuantityResults),
                                                           omit, omit)))
        {
            if (i == v_FieldCnt - 1) {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21: None of the LogMeasurementInfo matched the template");
            }
        }
        else {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 21");
            break;
        }
   }
 }//end of f_TC_8_1_6_1_2_1_NR5GC_TestBody



After Change:
	function f_TC_8_1_6_1_2_1_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var PLMN_Identity v_PlmnID_NRCell1 := f_NR_CellInfo_GetPLMN (nr_Cell1);
    var PLMN_Identity v_PlmnID_NRCell11 := f_NR_CellInfo_GetPLMN (nr_Cell11);
    var CellIdentity v_CellIdentity_NRCell1 := f_NR_CellInfo_GetCellIdentity(nr_Cell1); //Cell1
    var TrackingAreaCode v_TAC_NRCell1 := f_NR_CellInfo_GetTAC (nr_Cell1); //Cell1
    var integer i;
    var integer v_FieldCnt;
    var LogMeasInfo_r16 v_LogMeasInfo_r16;
    var template (present) MeasQuantityResults v_MeasQuantityResults;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var AbsoluteTimeInfo_r16 v_AbsoluteTimeInfo_r16 := f_MDT_GetAbsolute_TimeStamp();
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var ARFCN_ValueNR v_ARFCN_Neighbour := f_NR_CellInfo_GetFrequencySSB(nr_Cell11); //cell 11 arfcn
    var PhysCellId v_PhysCellIdNeighbour:= f_NR_CellInfo_GetPhysicalCellId(nr_Cell11); //cell 11 cellid id
    //@siclog "Step 1" siclog@
    // SS transmits a LoggedMeasurementConfiguration message
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                    cs_TimingInfo_Now,
                                    cs_38508_LoggedMeasurementConfiguration(v_PlmnID_NRCell1,
                                                                            v_AbsoluteTimeInfo_r16,
                                                                            cs_LoggedMeasurementConfiguration_reportType_Periodical)));

    
    //@siclog "Steps 2" siclog@
    // The SS transmits an RRCRelease message.
    f_NR_RRCRelease(nr_Cell1);

    //@siclog "Step 3" siclog@
    // Wait 5s to allow UE to activate logging.
    f_Delay (5.0);

    //@siclog "Step 4" siclog@
    //Set SS/PBCH EPRE levels acc. to row "T1"
    f_NR_SetCellPower (nr_Cell1, -85, -91);
    f_NR_SetCellPower (nr_Cell11, -78, -82);

    //@siclog "Step 5" siclog@
    //The UE transmits an RRCSetupRequest message on Cell 11.
    f_NR_RRC_SetupRequest_Def(nr_Cell11);

    //@siclog "Step 6" siclog@
    // The SS transmit an RRCSetup message.
    f_NR_RRCSetup_Def (nr_Cell11);

    //@siclog "Step 7" siclog@
    // Check: Does the UE include the IE logMeasAvailable in the RRCSetupComplete message?
    f_NR_RRCSetupComplete_Def(nr_Cell11,
                              cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def,-,-,-,cr_RRCSetupComplete_v1610_IEs(*,*,cr_UE_MeasurementsAvailable_r16(true_,omit,omit,omit,omit))));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.1.6.1.2.1 Step 7");

    //@siclog "Steps 8-10" siclog@
    // Steps 4 to 6a1 of generic test procedure in TS 38.508-1[4] subclause 4.9.5.2.2-1 are performed on Cell 11.
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11, v_GMM_MobilityInfo);

    //@siclog "Steps 11-18" siclog@
    // Steps 1 to 8 of the generic procedure in TS 38.508 Table 4.5.4.2-3 are executed to successfully complete the service request procedure on Cell 11.
    f_NR_Paging_Def(nr_Cell11, RRC_IDLE);
    f_NR_RRC_SetupRequest_Def(nr_Cell11);
    f_NR_RRCSetup_Def(nr_Cell11);
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell11,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def,-,-,-,cr_RRCSetupComplete_v1610_IEs(*,*,cr_UE_MeasurementsAvailable_r16(true_,omit,omit,omit,omit))));
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell11, v_ReceivedMsg);

    //@siclog "Step 19" siclog@
    // The SS transmits a UEInformationRequest message on Cell 11.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell11,
                                    -,
                                    cs_38508_UEInformationRequest(tsc_NR_RRC_TI_Def, -, true_)));

    // siclog "Step 20" siclog@
    // Check: Does the UE transmit a UEInformationResponse message with the IEs absoluteTimeStamp, traceReference, traceRecordingSessionRef and tce-Id set to same value as received in the LoggedMeasurementConfiguration message in step 1; and a LogMeasInfoList with at least two  entries with serving cell idle mode measurements and where in at least one of the entries the IE measResultListNR include a neighbouring cell measurement of Cell 11; and  where the relativeTimeStamp is increased between the subsequent LogMeasInfoList entries by at least the value of configured loggingInterval in the received LoggedMeasurementConfiguration message in step 1.
    v_MeasQuantityResults := f_NR_Get_MeasQuantityResults();
    SRB.receive(car_NR_SRB_RrcPdu_IND(nr_Cell11, tsc_NR_RbId_SRB2, // received on SRB2
                                       cr_38508_UEInformationResponse(tsc_NR_RRC_TI_Def, -,-,
                                                                      cr_LogMeasReport_r16(v_AbsoluteTimeInfo_r16,
                                                                                           cr_TraceReference_r16(v_PlmnID_NRCell1),
                                                                                           '001A'O,
                                                                                           '05'O,
                                                                                           ?,
                                                                                           omit,omit,omit 
                                                                                           )))) -> value v_ReceivedAsp;

    v_FieldCnt := lengthof(v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.logMeasReport_r16.logMeasInfoList_r16);
    if (v_FieldCnt < 2) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 20: Not enough LogMeasurementInfo entries")
    }
    for (i:=0; i<v_FieldCnt; i:=i+1){
        v_LogMeasInfo_r16 := v_ReceivedAsp.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.logMeasReport_r16.logMeasInfoList_r16[i];
        if (not match(v_LogMeasInfo_r16, cr_LogMeasInfo_r16(-, cr_CGI_Info_Logging_r16(v_PlmnID_NRCell1, v_CellIdentity_NRCell1, v_TAC_NRCell1), // Serving cell infor is Cell1
                                                           cr_MeasResultServCell_r16(v_MeasQuantityResults),
                                                           {cr_MeasResultLogging2NR_r16(v_ARFCN_Neighbour, {cr_MeasResultLoggingNR_r16(v_PhysCellIdNeighbour)})}, omit))) // added cr_MeasResultLogging2NR_r16 to check the contents
        {
            if (i == v_FieldCnt - 1) {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21: None of the LogMeasurementInfo matched the template");
            }
        }
        else {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 21");
            break;
        }
   }
 }//end of f_TC_8_1_6_1_2_1_NR5GC_TestBody



